WinActor変数ファイル表示（HTAを使用したテキストファイル表示）

WinActorの変数出力ノードにて作成されたファイルを見やすくするHTAツールです。
検索が使えるのでデバッグ効率は格段に向上するはずです。
テキストファイルやCSVファイルなど他への応用へもどうぞ。
#1 WinActor前提
NTT提供ノード/デバッグ/変数出力ノードにて出力されたファイルを対象
#2コード
```
<html>
<head> 
<title>GOGO</title> 
<HTA:APPLICATION  
     APPLICATIONNAME="textファイル内容表示"
     SCROLL="yes"
     SINGLEINSTANCE="yes" 
>


<script language="VBScript"> 
' --- テキスト読み込み---
Sub Window_onLoad 
        window.resizeTo 700,800 
End Sub  

'--- テキスト読込 ---
Sub ReadText
  Dim ITEM(10000),VALU(10000)
  Set objFso = CreateObject("Scripting.FileSystemObject")
  
  '--- パス取得済確認 ---
  if document.getElementsByTagName("input")(0).value = "" then
    msgbox "ファイルを指定して下さい", 16, "エラー"
  
  Else
    Sleep 1000
    Set objFile = objFso.OpenTextFile(MyTextBox.Value, 1)
    cnt = 1
    
    '--- テキストファイル読込 ---
    Do while Not objFile.AtEndOfStream
      inpf = objFile.ReadLine
      '--- この文字列が出現したら読込リセット ---winctorのためのリセット
      If inpf = "-----ここの文字列は実際の文言に変更してください---" Then
        cnt = 1
      Else
      
        '--- 「=」を区切り文字にして読込---
        If InStr(inpf, "=") > 0 Then
          ITEM(cnt) = Left(inpf, InStr(inpf, "=") -1 )
          If Len(inpf) - InStr(inpf, "=") > 0 Then
            VALU(cnt) = Right(inpf, Len(inpf) - InStr(inpf, "="))
          Else
            VALU(cnt) = ""
          End If 
          cnt = cnt + 1
        End If
      End If
    Loop
    
    '--- 出力ファイルのヘッダ ---
    OUTP = "<span contenteditable><TABLE width=100% border><TR><TD>変数名</TD><TD>値</TD></TR>"
    
    '--- 出力レコード ---
    For i = 1 To cnt - 1
      OUTP = OUTP & "<TR><TD>" & ITEM(i) & "</TD><TD>" & VALU(i) & "</TD></TR>"
    Next 
    
    '--- 出力ファイルのフッタ ---
    For i = 1 to 5
      OUTP = OUTP & "<TR><TD></TD><TD></TD></TR>"
    Next
    OUTP = OUTP & "</TABLE></span>"
    TableArea.innerHtml = OUTP
    objFile.Close
  End if
  
  Set objFile = Nothing
  Set objFso = Nothing
End Sub


'--- テキスト出力---
Sub WriteText
  
  if Msgbox("上書き更新しますがよろしいですか", 49, "ファイル更新（上書きします）") = "1" then
    
    Sleep 1000
    Set objFso = CreateObject("Scripting.FileSystemObject")
    
    '--- パス取得済確認 ---
    if document.getElementsByTagName("input")(0).value = "" then
      msgbox "ファイルを指定して下さい", 16, "エラー"
    Else
      If document.getElementsByTagName("TABLE").length = 0 then
        msgbox "読込が完了していないと出力はできません", 16, "エラー"
      Else
        Set objFile = objFso.OpenTextFile(MyTextBox.Value, 2, True)
        Set TABLE = document.getElementsByTagName("TABLE")(0)
        
        i = 1
        While TABLE.rows(i).cells(0).innerText <> ""
          objFile.WriteLine TABLE.rows(i).cells(0).innerText & "=" & TABLE.rows(i).cells(1).innerText
          i = i + 1
        Wend
        objFile.Close
      End if
    End if
    
    Set objFile = Nothing
    Set objFso = Nothing
  
  End if
End Sub

Sub Sleep(millisec)
  Dim dstart
  dstart = timer * 1000
  Do while True
    if timer * 1000 >= dstart + millisec then Exit Do
  Loop
End Sub


</script></head>


<body> 
<input name="MyTextBox" type="file" size=100 id="selfile"><p>
ファイルパスを指定し、読込ボタンを押下して下さい。<br>
内容に修正がある場合は修正後に出力ボタンを押下して下さい。<p>
<input type="button" value="読込" name="run_button" onClick="ReadText">　　
<input type="button" value="出力" name="run_button2" onClick="WriteText">　（※出力は上書きです。読込後しか上書き出来ません）
<p>
<p>
<span id=TableArea></span>
<p>


</body> 
</html>

```
#3コード補足
・xxxx.hta　としてファイルに保存してください。
・対象としている変数が格納されたファイルは「追加型」でノード実行時に元のファイルに今回分が追加されています。
・このツールでは最新のデータのみを対象としています。
・「前回データ」と「今回データ」の区切りは「-----・・・・」で区切られています。
・この区切り文字はご自分のシナリオバージョン合わせて変更してください。
#4実行
・実行はダブルクリックで実行。
・実行時「時間がかかります・・・」と表示されますが、500MBで1分程度です。

・ファイル更新もできるようにしています。変数設定ノードを利用することでシナリオの途中から変数値を自由に変更できるので、注意しながら使用してもらうとさらに開発効率があがります。